CIVIL ENGINEERS
POCKET COMPUTER MONTHLY

BEAMS - SIMPLE AND CONTINUOUS - PART II

In this issue we will continue to examine beams and will provide PC-1/PC-2 programs for shears, moments and reactions.

Using the programs provided herein and in last month's issue we will analyze a continuous four span beam for four loading patterns, including shear and moment diagram.

PROGRAM - BEAM ANALYSIS-SHEAR/ MOMENT/REACTIONS

Program solves for left and right reactions; also solves for shears and moments at equally-spaced intervals along the span. The designer may then plot shear and moment diagrams to determine:

- Max Negative Moments
- Max Positive Moment
- Locations of Points of Zero Moment (inflection)
- Location of Point of Zero Shear



FREE - BODY DIAGRAM

Program Listing - PC-1

This PC-1 program accomodates uniform loads and concentrated loads.

```
 1Ø  "A" PRINT "*BEAM ANALYSIS*":
     CLEAR:PRINT " "
1ØØ  INPUT "SPAN (L),FT= ";P
11Ø  INPUT "LEFT END MOM (FT-K)=
     ";M
12Ø  INPUT "RIGHT END MOM(FT-K)=";O
```

Max. No. of Unif. Loads = 3

```
14Ø  INPUT "NO. OF UNIF LDS=";N:IF
     N>3 GOSUB 6ØØ:GOTO 14Ø
15Ø  Q=Ø:R=Ø:IF N=Ø THEN 25Ø
17Ø  FOR J=1 TO N:K=J+3:L=K+3:INPUT
     "ENTER UNIT LD(K/FT)=";A(J)
2ØØ  INPUT "ENTER DIST. L(A),FT=";
     A(K):IF A(K)>P GOSUB 6ØØ:GOTO 2ØØ
21Ø  INPUT "ENTER DIST. L(B),FT=";A(L):
     IF A(L)> PGOSUB 6ØØ:GOTO 21Ø
22Ø  Q=Q+A(J)*(A(L)-A(K)): R=R+A(J)*
     (A(L)-A(K))*(A(K)+A(L))/2:NEXT J
25Ø  INPUT "NO. OF CONC. LDS=";T:IF
     T> 8 GOSUB 6ØØ:GOTO 25Ø
```

Max. No. of Conc. Loads =8

```
26Ø  IF T=Ø THEN 32Ø
27Ø  FOR J=27 TO T+26:K=J+8:INPUT
     "ENTER LOAD, KIPS=";A(J)
3ØØ  INPUT "ENTER DIST. D,FT=";A(K):IF
     A(K)> P GOSUB 6ØØ:GOTO 3ØØ
31Ø  Q=Q+A(J):R=R+A(J)*A(K):NEXT J
32Ø  USING "######.#":PRINT P:PRINT
     M:PRINT O:USING:PRINT N:PRINT T
325  PRINT " ":PRINT " **OUTPUT**":
     PRINT " ":R=(R+M+O)/P: Q=Q-R
34Ø  PRINT "REACTIONS (L/R)=":PRINT Q,R
```

Max. No. of increments = 1Ø

```
36Ø  INPUT "NO. OF INCR=";S:IF
     S>1Ø GOSUB 6ØØ:GOTO 36Ø
37Ø  FOR J=43 TO 43+S:K=J+S:X=(J-43)*
     P/S:A(J)=M+X*Q:A(K)=Q
```

Note: Calculates shear and moment at each increment, for uniform loads.

```
41Ø  FOR V=1 TO N:U=V+3:W=U+3:IF
     X<=A(U) THEN 5ØØ
44Ø  IF X>=A(W) THEN 48Ø
45Ø  A(J)=A(J)-A(V)*(X-A(U))^2/2:
     A(K)=A(K)-A(V)*(X-A(U)):GOTO 5ØØ
48Ø  A(J)=A(J)-A(V)*(A(W)-A(U))*(X-(A(W)+
     A(U))/2:A(K)=A(K)-A(V)*(A(W)-A(U))
5ØØ  NEXT V
```

Calculates and prints shear and moment at each increment, for conc. loads.

```
51Ø  FOR V=1 TO T
52Ø  U=V+26:W=V+34:IF X+ .ØØØØ1
     <A(W) THEN 55Ø
54Ø  A(J)=A(J)-A(U)*(X-A(W)):
     A(K)=A(K)-A(U)
55Ø  NEXT V
57Ø  R=J-43:BEEP 1:PRINT "INCR
     NO. ";USING;R:USING "####.#":
     PRINT "X=":PRINT X
575  PRINT "M=":PRINT USING
     "########.#"; A(J):PRINT "V=":
     PRINT A(K):PRINT " ":NEXT J:END
6ØØ  BEEP 4:PAUSE "ERROR":RETURN
```

Program Listing - PC-2

This PC-2 program accomodates uniform loads, concentrated loads, and ramp loads (SDL & SDR)

```
 1Ø  "A" LPRINT "**BEAM ANALYSIS**":
     CLEAR:LF 2:WAIT 1ØØ:USING
 2Ø  DIM A1(5Ø), A2(5Ø), A3(5Ø),
     B1(5Ø), B2(5Ø), C1(5Ø), C2(5Ø),
     C3(5Ø)
 3Ø  DIM D1(5Ø), D2(5Ø), D3(5Ø)
1ØØ  INPUT "SPAN(FT)= ";P:LPRINT
     "SPAN,FT= ":LPRINT P
11Ø  INPUT "LEFT END MOM(FT-K)=";M
115  IF M<>Ø LPRINT "L.END MOM
     (FT-K)= ":LPRINT M
12Ø  INPUT "RIGHT END MOM(FT-K)= ";O
125  IF O<>Ø LPRINT "R. END
     MOM(FT-K)= ":LPRINT O
14Ø  INPUT "NO. OF UNIF. LDS= ";N
15Ø  IF N>Ø LPRINT "NO. OF UNIF.
     LDS = ":LPRINT N:GOTO 17Ø
154  REM
155  REM MAX. NO. OF UNIF. LOADS=51
156  REM
16Ø  GOTO 25Ø
17Ø  FOR J=1 TO N:INPUT "ENTER
     UNIT LD (K/FT)=";A1(J)
2ØØ  INPUT "ENTER DIST. L(A),FT.=";A2(J)
2Ø5    IF A2(J)>P PRINT "ERROR":GOTO
       2ØØ
21Ø    INPUT "ENTER DIST. L(B),FT.=";A3(J)
215  IF A3(J)>P PRINT "ERROR":
     GOTO 21Ø
22Ø  Q=Q+A1(J)*(A3(J)-A2(J)):R=R+A1(J)*
     (A3(J)-A2(J))*(A2(J)+A3(J))/2
23Ø  LPRINT "UNIT LOAD # ";J:
     LPRINT " U. LD (K/FT)= ",
     A1(J):LPRINT " DIST. L(A),FT=",A2(J)
24Ø  LPRINT " DIST.L(B),FT=",A3(J):NEXT J
25Ø  INPUT "NO. OF CONC. LDS=
     ";T:IF T=Ø THEN 38Ø
26Ø  LPRINT "NO. OF CONC. LDS=",T
264  REM
265  REM MAX. NO. OF CONC. LOADS=51
266  REM
27Ø  FOR J=1 TO T: INPUT "ENTER
     LOAD, KIPS= ";B1(J)
28Ø  INPUT "ENTER DIST. D, FT=";B2(J)
285  IF B2(J)>P PRINT "ERROR":GOTO 28Ø
29Ø  LPRINT "CONC. LOAD # ";J:
     LPRINT " C.LD, K = ",B1(J):
     LPRINT " DIST. D = ",B2(J)
31Ø  Q=Q+B1(J):R=R+B1(J)*B2(J):NEXT J
38Ø  INPUT "NO. OF RAMP(SDL)
     LOADS=";NN: IF NN=Ø THEN 58Ø
389  REM
39Ø  REM MAX. NO. OF RAMP (SDL)
     LOADS=51
391  REM
4ØØ  FOR J=1 TO NN:INPUT "ENTER
     DIST. L(A),FT= ";C2(J):IF C2(J)>P
     PRINT "ERROR":GOTO 4ØØ
41Ø  INPUT "ENTER DIST. L(B),FT=";
     C3(J):IF C3(J)>P PRINT "ERROR":
     GOTO 41Ø
43Ø  INPUT "ENTER MAX. UNIT
     LD(K/FT)=";C1(J)
44Ø  Q=Q+C1(J)*(C3(J)-C2(J))/2
45Ø  R=R+C1(J)*(C3(J)-C2(J))/2*(C2(J)+
     (C3(J)-C2(J))*2/3)
46Ø  LPRINT "SDL LOAD # ",J:LPRINT
     " MAX. LD(K/FT)=",C1(J):LPRINT
     " DIST. L(A),FT=",C2(J)
47Ø  LPRINT "DIST.L(B),FT=",C3(J):NEXT J
58Ø  INPUT "NO. OF RAMPS (SDR)
     LOADS="MM:IF MM=Ø THEN 68Ø
589  REM
59Ø  REM MAX. NO. OF RAMP (SDR)
     LOADS=51
591  REM
6ØØ  FOR J=1 TO MM: INPUT "ENTER
     DIST. L(A),FT=";D2(J):IF D2(J)>P
     PRINT "ERROR":GOTO 6ØØ
```

```
610  INPUT "ENTER DIST. L(B),FT=
     ";D3(J):IF D3(J)>P PRINT "ERROR":
     GOTO 610
630  INPUT "ENTER MAX. UNIT
     LD(K/FT)=";D1(J)
640  Q=Q+D1(J)*(D3(J)-D2(J))/2
650  R=R+D1(J)*(D3(J)-D2(J))/2*(D2(J)+
     (D3(J)-D2(J))/3)
660  LPRINT "SDR LOAD # ";J:
     LPRINT " MAX LD (K/FT)=",D1(J):
     LPRINT " DIST. L(A),FT=",D2(J)
670  LPRINT " DIST. L(B),FT=",D3(J):
     NEXT J
674  REM
675  REM SELECT NO. OF INCREMENTS
     IN SHEAR & MOMENT DIAGRAMS
676  REM
680  INPUT "NO. OF INCR=";S
690  LPRINT "NO. OF INCR=  ",S
699  REM
700  REM PGM CALCULATES &
     PRINTS OUTPUT
701  REM
725  LF 2:LPRINT "**OUTPUT**":
     LPRINT: R=(R+M+O)/P:Q=Q-R:
     USING "########.##"
740  LPRINT "REACTIONS (L/R),K:":
     LPRINT INT (Q*100+.5)/100,
     INT (R*100+.5)/100:LF 2
870  FOR J=0 TO S
880  X=J*P/S
890  E1=M+X*Q
895  E2=Q
897  REM
898  REM FOR UNIF LOADS
899  REM
900  FOR V=1 TO N
910  IF X<=A2(V)THEN 950
920  IF X<A3(V) THEN 940
930  E1=E1-A1(V)*(A3(V)-A2(V))*(X
     -(A3(V)+A2(V))/2)
933  E2=E2-A1(V)*(A3(V)-A2(V))
935  GOTO 950
940  E1=E1-A1(V)*(X-A2(V))^2/2:E2=
     E2-A1(V)*(X-A2(V))
950  NEXT V
959  REM
960  REM FOR CONC LOADS
961  REM
1000 FOR V=1 TO T
1010 IF X<=B2(V) THEN 1050
1040 E1=E1-B1(V)*(X-B2(V)):E2=E2-B1(V)
1050 NEXT V
1059 REM
1060 REM FOR RAMP SDL LOADS
1061 REM
1090 IF NN=0 THEN 1190
1100 FOR V=1 TO NN
1105 Z=C1(V)/(C3(V)-C2(V))
1110 IF X<=C2(V) THEN 1170
1120 IF X<C3(V) THEN 1160
1140 E1=E1-(X-(C3(V)-C2(V))*2/3-C2(V))*
     (C3(V)-C2(V))^2/2*Z
1142 E2=E2-Z*(C3(V)-C2(V))^2/2
1150 GOTO 1170
1160 E1=E1-(X-C2(V))^3/6*Z:E2=E2-
     (X-C2(V))^2/2*Z
1170 NEXT V
1179 REM
1180 REM FOR RAMP SDR LOADS
1181 REM
1190 IF MM=0 THEN 1280
1200 FOR V=1 TO MM
1205 Z=D1(V)/(D3(V)-D2(V))
1210 IF X<=D2(V) THEN 1270
1220 IF X<D3(V) THEN 1260
1230 E1=E1-(X-(D3(V)-D2(V))/3-D2(V))*(D3
     (V)-D2(V))^2/2*Z
1233 E2=E2-Z*(D3(V)-D2(V))^2/2
1240 GOTO 1270
1260 E1=E1-(X-D2(V))^2*D1(V)/2+
     (X-D2(V))^3/6*Z
1265 E2=E2-(X-D2(V))*D1(V)+(X-D2(V))
     ^2/2*Z
1270 NEXT V
1280 USING:LPRINT "INCREMENT #
     ";J
1290 LPRINT "X(FT)=  ";INT (100*
     X+.5)/100
1295 USING "########.##"
1300 LPRINT "MOMENT(FT-K)=",
     INT (E1*100+.5)/100
1310 LPRINT "SHEAR(K)= ",INT
     (100*E2+.5)/100:LF 4
1320 NEXT J
```

Note (PC-2 Program):

The limit on number of loads of various kinds is defined by the DIM statement (Lines 20 and 30). We arbitrarily selected DIM 50 for all parameters. You may exchange this as you wish. Note that there is no limit in this program on the number of increments.

HELPFUL HINTS FOR PC-2/PC-1500

DEF: GOTO: GOSUB

The letters on the keyboard, with the exception of the top ("QWERTY") row, may be defined. Where a line begins with "A" (e.g., 10 "A" X=1) the program may be started on that line by pressing DEF A. Suppose the user wishes to have the option of starting a program with all variables cleared to zero and all strings cleared to null. The following would accomplish thus:

```
10 "A" CLEAR
20 "B"
```

Pressing DEF A would start the program cleared, whereas DEF B would restart the program with prior values of variables and strings retained in memory.

The list of definable characters includes those in the second and third rows, including = but excluding (and) and ↑ . Also, numbers are excluded.

For example, the following program will print out a series of) until ON is pressed:

```
10 "=" LPRINT ")":GOTO 10
Press DEF =
```

also,

```
10 "=" LPRINT ")":GOTO "="
```

also,

```
 5 "A"
10 "SALLY" LPRINT ")":GOTO
    "SALLY"
```

will produce the same effect when DEF A is pressed.

Also

```
 5 "A" X$="SALLY"
10 "SALLY" LPRINT ")"
15 GOTO X$
```

will print out a list of)'s until ON is pressed to break the action.

The same may be used as a GOSUB. For example:

```
 5 "A" X$="SALLY"
10 GOSUB X$:GOTO 10
15 "SALLY" LPRINT ")":RETURN
```

produces the same effect.

IF-THEN

The logical IF command is a control statement which evaluates given conditions and either executes the command or moves execution down to the next line.

In contrast with other forms of BASIC, PC-2's BASIC does not necessarily require the word "THEN".

After the comparison is made (e.g., IF X=1) the following are illustrations of BASIC statements/commands which may be inserted:

```
THEN
GOSUB
LET
PRINT OR LPRINT OR PAUSE
INPUT
READ
ON
etc.
```

The word THEN is optional except in the form: 10 IF X=1 THEN 100 which sends the pointer to Line 100 if X=1. Alternate forms are:

```
10 IF X=1 GOTO 100
10 IF X=1 THEN GO TO 100
```

The following form imposes two tests. We will write it in two acceptable forms:

```
10 IF X=1 AND Y=1 THEN 100
10 IF (X=1)*(Y=1) THEN 100
```

similarly

```
10 IF X=1 OR Y=1 THEN 100
10 IF (X=1)+(Y=1) THEN 100
```

This gets more complicated in the following form:

```
10 IF ((W=1)+(X=1)+(Y=1)+(Z=1))
    THEN 100
```

which means that if W=1 or X=1 or Y=1 or Z=1, then GOTO 100

More complicated yet:

```
10 IF    ((V=0)*((W=1)+(X=1)+(Y=1)+
    (Z=1))) THEN 100
```

which means that if V=0 and if either W=1 or X=1 or Y=1 or Z=1 then go to 100.

It is also possible to assign a logical value as follows:

```
1Ø V=(V>W)
```

then the logical IF takes the following form:

```
2Ø IF V THEN 1ØØ,which means
2Ø IF V>W THEN 1ØØ
```

The "default" of this form is

```
V<>Ø
```

that is

```
2Ø IF V<>Ø THEN 1ØØ
```

may be written as

```
2Ø IF V THEN 1ØØ
```

It is permissible to create a new variable for this purpose, thus:

```
1Ø X=(V>W)
2Ø IF X THEN 1ØØ which means
2Ø IF V>W THEN 1ØØ
```

ERRATA

April Issue:

On Page 6, Second column. The program as shown, will run on PC-2. However, in order to run on PC-1, make the following changes:

Line 1ØØ Eliminate WAIT 5ØØ:
Line 11Ø Change AND to IF
Note that in Lines 112, 114, 116, 118 the "0" should be a zero (Ø)

Line 21Ø-Page 7 Change AND to IF

We note that having loaded this program into PC-1 and having pressed SHIFT A in the DEF Mode, we get Error Message 4 (Inadequate Memory). To overcome this problem, eliminate GCP program lines 5Ø1 through 54Ø. Replace with 5Ø1 END. When the program has ceased running, a prompt will appear in the display. Press A ENTER, to display the output, that is, the interaction number.

May Issue:

In Problem 5, Page 7 we find that the example run for Span #4 was in error. FEM left should have been -27.00, FEM right should have been 16.48.

We offer Cassette Tapes with documentation at $20 each for:

1) Statistics-Confidence Program, PC-1 or PC-2
2) General Computation Program with or without printer, PC-1 or PC-2
3) Wood Column; Wood Foundation Wall; Wood Members - Combined Axial and Bending - PC-1 or PC-2
4) Steel Column and Steel Beam-Column, PC-2
5) Hazen-Williams and Chezy Manning Flow Formulas, PC-1 or PC-2
6) Walls Retaining Drained Sand, Active Case, PC-1 or PC-2
7) Pratt Truss, PC-2 (requires 8K Module)
8) Beams - Fixed End Moment and Four Span Moment Distribution (two versions), PC-1 or PC-2
9) Beam Analysis - Shear/Moment/Reactions, PC-1 or PC-2
10) Accounts Receivable, PC-2. This program prints detailed invoices, including aging. Prepares summary of business during the time period. Suitable for up to five hourly categories of charge rates. Requires 8K module.

Note: Civil Engineers Pocket Computer Monthly supports Radio Shack's PC-1 and PC-2 (Sharp PC-1211 and PC-1500). We believe our software will be helpful to civil engineers who have other equipment.

The software provided in this issue is solely for educational and experimental purposes. It is supplied "as-is" without warranty of any kind. We do not assume any liability for any direct, indirect, incidental or consequential damages relating to the use or application of the programs or information contained herein.

WORKED OUT EXAMPLES

Continuous Beam, 4-Span, Uniform Moment of Inertia

Free Body Diagram:



Loadings are in kips or in kips/ft., as applicable.

Use Fixed End Moment Program (PC-2), May Issue, to solve for FEM's

Computer Printouts, FEM:

Span #1

```
****SUM OF FEM****

FEM, LEFT=  -212.41
FEM, RGHT=   191.01
```

Span #2

```
****SUM OF FEM****

FEM, LEFT=  -282.72
FEM, RGHT=   256.28
```

Span #3

```
****SUM OF FEM****

FEM, LEFT=  -242.99
FEM, RGHT=   301.31
```

Span #4

```
****SUM OF FEM****

FEM, LEFT=  -222.97
FEM, RGHT=   166.26
```

We then used the PC-1 program, May Issue, to carry out Moment Distribution. Eight cycles were specified. Note that two versions of the program were provided in the May Issue; we are using the version requiring input of FEM's. Note that the left end moments in each span are counter-clockwise and therefore must be entered as negative numbers.

```
END MOMENTS(L/R)

SPAN# 1.
          0.000
        291.306

SPAN# 2.
       -291.306
        247.603

SPAN# 3.
       -247.603
        302.029

SPAN# 4.
       -302.029
          0.000
```

Then we ran the enclosed program for each of the four spans and received the following output:

SUMMARY

Span #	Span, ft.	Reactions, K Left	Right	Maximum Positive* Moment, Ft-K
1	30	24.16	39.84	193.13 at X=12'
2	42	37.21	31.79	134.80 at X=21'
3	38	28.65	40.05	173.82 at X=19'
4	28	51.00	17.90	149.23 at X=16.8'

*In this context, "positive" moment is defined as "tension in bottom of members".

We have selected Span #3 to illustrate the computer output:

```
**BEAM  ANALYSIS**

SPAN, FT=
                  38
L.END MOM(FT-K)=
           -247.603
R.END MOM(FT-K)=
            302.029
NO.OF UNIF.LDS=
                   1
UNIT LOAD #   1
  U.LD(K/FT)=
                 0.4
  DIST.L(A), FT.=
                   0
  DIST.L(B), FT.=
                  38
NO.OF CONC.LDS=
                   1
CONC.LOAD #   1
  C.LD(K), FT=
                  13
  DIST.D     =
                  19
SDL LOAD #
                   1
  MAX.LD(K/FT)=
                   3
  DIST.L(A), FT=
                   5
  DIST.L(B), FT=
                  32
NO.OF INCR=     10
```

```
**OUTPUT**

REACTIONS(L/R), K:
                28.65
                40.05

INCREMENT #   0
X(FT)=   0
MOMENT(FT-K)=
             -247.60
SHEAR(K)=
                28.65

INCREMENT #   1
X(FT)=   3.8
MOMENT(FT-K)=
             -141.60
SHEAR(K)=
                27.13

INCREMENT #   2
X(FT)=   7.6
MOMENT(FT-K)=
              -41.71
SHEAR(K)=
                25.24
```

```
INCREMENT #   3
X(FT)=   11.4
MOMENT(FT-K)=
                48.21
SHEAR(K)=
                21.82

INCREMENT #   4
X(FT)=   15.2
MOMENT(FT-K)=
               122.09
SHEAR(K)=
                16.79

INCREMENT #   5
X(FT)=   19
MOMENT(FT-K)=
               173.82
SHEAR(K)=
                10.17

INCREMENT #   6
X(FT)=   22.8
MOMENT(FT-K)=
               147.91
SHEAR(K)=
               -11.07
```

```
INCREMENT #  7
X(FT)=  26.6
MOMENT(FT-K)=
              87.67
SHEAR(K)=
             -20.91


INCREMENT #  8
X(FT)=  30.4
MOMENT(FT-K)=
             -13.00
SHEAR(K)=
             -32.35


INCREMENT #  9
X(FT)=  34.2
MOMENT(FT-K)=
            -152.74
SHEAR(K)=
             -38.53


INCREMENT #  10
X(FT)=  38
MOMENT(FT-K)=
            -302.03
SHEAR(K)=
             -40.05
```


SPAN #3
19 FT.
13K
32 FT.
3 K/FT
5 FT
Ø.4 K/FT
R_L = 28.65 K
R_R = 4Ø.Ø5 K
38 FT. SPAN
FREE-BODY DIAGRAM
SHEAR
MOMENT
4Ø
3Ø
2Ø
1Ø
Ø
-1Ø
-2Ø
-3Ø
-4Ø
-5Ø
-6Ø
-7Ø
24Ø
2ØØ
16Ø
12Ø
8Ø
4Ø
Ø
-4Ø
-8Ø
-12Ø
-16Ø
-2ØØ
-24Ø
-28Ø
-32Ø
173.82 FT. K.
MOMENT
SHEAR
247.6Ø
3Ø2.Ø3
SHEAR & MOMENT DIAGRAM